MANIPULATIONS ALGÉBRIQUES

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Maple est plus qu'une calculatrice. Il permet de faire des mathématiques symboliques, autrement dit manipuler des expressions dans lesquelles figurent des variables - qui ne représentent pas nécessairement des nombres - afin d'en étudier les propriétés. Ceci étant dit, Maple ne fait pas des mathématiques à la place de l'utilisateur. Et celui ci risque fort des déconvenues s'il compte se reposer totalement sur les capacités du logiciel pour atteindre un résultat. Maple demande à être guidé dans les tâches de simplification des résultats, ce qui suppose que l'utilisateur connaît suffisamment la théorie mathématique et les capacités du logiciel pour jouer utilement le rôle de mentor. Ce chapitre est consacré aux manipulations algébriques élémentaires et il sera utile en pratique quand on a besoin de simplifier une expression algébrique.

Après avoir rappelé quelles sont les manipulations simples qu'effectue automatiquement Maple (section 1), ce chapitre traite de la manipulation des expressions polynômiales d'une part (section 2) et rationnelles d'autre part (section 3). Il s'agir d'apprivoiser des techniques algébriques telles que développer, factoriser, réduire au même dénominateur, simplifier, etc..

Manipulations automatiques

Avant de guider Maple, prenons la mesure de ce que qu'il fait sans qu'on lui demande quoi que ce soit.

Dans le chapitre "Dialoguer et calculer avec Maple", on a vu que Maple exprime sous forme irréductible les nombres rationnels et leur attribue automatiquement le type fraction ou integer.

```
> restart;
  152/46;whattype(%);
  98/49;whattype(%);
```

$$\frac{76}{23}$$

$$\mathit{fraction}$$

$$2$$

$$\mathit{integer}$$

Maple réduit les sommes, produits et puissances de **nombres** rationnels du type fraction :

```
> 152/46+98/49;#somme de fractions
  152/46-98/49;#différence de fractions
  152/46*98/49;#produit de fractions
  (152/46)^3;#puissance d'une fraction
```

$$\frac{122}{23}$$

$$\frac{30}{23}$$

$$\frac{152}{23}$$

$$\frac{438976}{12167}$$

Mais il se contente du minimum syndical sur des fractions faisant intervenir des symboles :

```
> 152/a+b/49;
  a/x*x/y;#cas où il y a simplification automatique
  a/b*c/d;
```

$$\frac{152}{a}+\frac{1}{49}\,b$$

$$\frac{a}{y}$$

$$\frac{a\,c}{b\,d}$$

Maple effectue la distribution du produit d'un **nombre** sur une somme de termes :

```
> 4*(a+b+c);#distribution automatique
  a*(2+f);#pas de distribution
  (x+y)*(x-a);#pas de distribution
```

$$4\,a+4\,b+4\,c$$

$$a\,(2+f)$$

$$(x+y)\,(x-a)$$

Il place en tête la constante numérique dans un produit de facteurs :

```
> a*b*6*c*5;
```

$$30\,a\,b\,c$$

Il regroupe des termes syntaxiquement identiques dans une somme ou un produit :

```
> 4*a*b+4*a*3+4*a;
  a*b*2*b*b*a;
```

$$4\,a\,b+16\,a$$

$$2\,a^2\,b^3$$

Il simplifie des facteurs syntaxiquement identiques dans un quotient, mais c'est assez aléatoire :

```
> (4*a*b)/(3*a);#cas où il y a simplification
  (x+y)(2*x-a)/(2*x-a)(x^2+y^2);#cas où il n'y a pas de
  simplification
```

$$\frac{4}{3}\,b$$

$$\frac{x(2\,x-a)+y(2\,x-a)}{2\,x(x^2+y^2)-a(x^2+y^2)}$$

Il simplifie quelques valeurs numériques remarquables telles que :

```
> ln(1);cos(Pi/3);exp(x-x);
```

$$0$$

$$\frac{1}{2}$$

$$1$$

Il réécrit quelques expressions en expressions équivalentes qu'il a rencontrées au préalable :

```
> 1+x^2+x*y+5*x+0.2*a*y;#expression polynômiale
  x^2+x*y+5*x+0.2*a*y+1;#expression polynômiale équivalente
```

$$1+x^2+x\,y+5\,x+0.2\,a\,y$$

$$1+x^2+x\,y+5\,x+0.2\,a\,y$$

Il simplifie quelques évidences :

```
> sin(arcsin(x));
  (sqrt(a))^2;
```

$$x$$

$$a$$

Là s'arréte le champ des manipulations automatiques. Il semble restreint mais la règle de prudence l'emporte sur toute autre considération. On pourra même être surpris par des réponses excessivement - mais volontairement - prudentes. Par exemples :

```
> (x^3-y^3)/(x-y);#Maple ne voit pas la simplification évidente
```

$$\frac{x^3-y^3}{x-y}$$

```
> (a-b)*(a+b);#Maple semble ignorer les identités remarquables
```

$$(a-b)\,(a+b)$$

```
> cos(x)^2+sin(x)^2;#Maple n'applique pas des propriétés
  élémentaires
```

$$\cos(x)^2+\sin(x)^2$$

```
> log(a*b)-log(a)-log(b);#Même remarque que précédemment
```

$$\ln(a\,b)-\ln(a)-\ln(b)$$

Ainsi, Maple semble au premier abord inutile puisque les capacités de traitement automatique d'expressions algébriques aussi simples que des polynômes ou des quotients de polynômes sont assez minces. Heureusement, il n'en est rien comme on va le voir dans les sections suivantes.

Manipulations d'expressions polynômiales

Dans la plupart des cas, on considère dans cette section des expressions polynômiales de la variable x sous la forme d'une somme de monômes $a_0+a_1\,t+a_2\,x^2+...+a_n\,x^n=\sum_{k=0}^{n} a_k\,x^k$ ou sous la forme d'un produits de facteurs $a_0\,(x-a_1)^{n_1}\,...\,(x-a_p)^{n_p}$. Mais les manipulations étudiées s'étendent sans difficulté à des polynômes de plusieurs variables.

Les manipulations courantes sont développer et factoriser, mais il peut aussi être utile de regrouper des termes ou de les ordonner suivant les puissances décroissantes.

Développer un polynôme : `expand()`

Soit `Poly` une expression polynômiale donnée sous forme d'un produits de facteurs.
L'instruction `expand(Poly)` renvoie une somme de produits de facteurs élémentaires.

```
> expand((x+1)*(x-2)^2);
  expand((t+a)*(t+b)^2);
```

$$x^3 - 3x^2 + 4$$
$$t^3 + 2t^2 b + t b^2 + a t^2 + 2 a t b + a b^2$$

Par définition, un développement fait jouer la distributivité de la multiplication sur une somme. La commande **expand** est basée sur ce principe et on peut donc indiquer en option le ou les polynômes dont on désire conserver l'expression.

```
> expand((x+1)*(x-2)^2,x+1);
  expand((t+a)*(t+b)^2,t+b);
```

$$(x+1)\, x^2 - 4\,(x+1)\, x + 4x + 4$$
$$(t+b)^2\, t + (t+b)^2\, a$$

Factoriser un polynôme : factor( )

Soit **Poly** une expression polynômiale donnée sous forme d'une somme de monômes. L'instruction **factor(Poly)** renvoie normalement un produit de facteurs.

```
> factor(x^8-1);#exemple 1
  factor(x^2-1);#exemple 2
  factor(x^2-2);#exemple 3
```

$$(x-1)\,(x+1)\,(x^2+1)\,(x^4+1)$$
$$(x-1)\,(x+1)$$
$$x^2 - 2$$

Le résultat de l'exemple 3 est décevant. Il se comprend en se référant à la logique mathématique retenue par les programmeurs : l'instruction **factor** factorise un polynôme sur le sous-corps (des complexes) engendré par les coefficients. Dans le polynôme $x^2 - 2$, les coefficients sont des entiers naturels et Maple cherche à résoudre $x^2 - 2 = 0$ dans le corps des rationnels. Comme la solution n'existe pas ($\sqrt{2}$ n'est pas un nombre rétionnel), il renvoie en écho le polynôme initial. Ce problème est corrigé en obligeant Maple à travailler dans le corps des réels. Une première possibilité est de contraindre les coefficients à prendre le type float en ajoutant un point à au moins un d'entre eux:

```
> factor(1.*x^2-2.);
```

$$(x + 1.414213562)\,(x - 1.414213562)$$

Mais on préférerait voir apparaître $\sqrt{2}$. Il est possible d'indiquer en second argument de **factor** que le champ des coefficients doit être étendu et comprendre $\sqrt{2}$.

```
> factor(x^2-2,sqrt(2));
```

$$-\left(-x+\sqrt{2}\right)\left(x+\sqrt{2}\right)$$

Bien entendu, il s'agit là d'un artifice qui fonctionne ici parce que tout le monde connaît un minimum d'identités remarquables. Dans le cas général, on passera en second argument l'instruction **RootOf(Poly)** :

```
> factor(x^2-7,RootOf(x^2-7));
```

$$-\left(x + \mathit{RootOf}\left(\_Z^2 - 7\right)\right)\left(-x + \mathit{RootOf}\left(\_Z^2 - 7\right)\right)$$

La réponse est juste, mais quelque peu énigmatique. En réalité, Maple factorise correctement, mais ne donne pas la racine explicite de l'équation $x^2 - 7 = 0$. Dans ce cas, où tous les coefficients de l'équation sont des nombres, on peut demander une valeur décimale approchée des racines par la commande **evalf** :

```
> evalf(RootOf(x^2-7));
```

$$2.645751311$$

Ceci dit, on ne fait toujours pas apparaître $\sqrt{7}$. La commande **convert** avec en option **radical** résout ce problème :

```
> P:=factor(x^2-7,RootOf(x^2-7));
  convert(P,radical);
```

$$P := -(x + RootOf(\_Z^2 - 7))\,(-x + RootOf(\_Z^2 - 7))$$

$$-(x + \sqrt{7})\,(-x + \sqrt{7})$$

Regroupement de termes : collect( )

Pour regrouper dans un polynôme les termes de même puissance, on utilise l'instruction **collect** qui a deux arguments : le premier est le polynôme étudié; le second précise la variable de regroupement. L'exemple qui suit montre bien la différence entre un développement et un regroupement de termes.

```
> expand((x-a)*(x-b)*(x-c));
  collect((x-a)*(x-b)*(x-c),x);
```

$$x^3 - x^2 c - x^2 b + x b c - a x^2 + a x c + a b x - a b c$$

$$x^3 + (-a - b - c)\,x^2 + (a b - (-a - b)\,c)\,x - a b c$$

Les programmeurs ont voulu que **collect** fasse préalablement appel à **expand**, de sorte qu'il n'est pas utile de développer un polynôme puis de regrouper les termes.

Le deuxième argument peut aussi être un *n*-uple de variables ordonnées, qui se forme en entourant de crochets les variables considérées.

```
> collect((x-a)*(x-b)*(x-c),x);#regroupement suivant x
  collect((x-a)*(x-b)*(x-c),a);#regroupement suivant a
  collect((x-a)*(x-b)*(x-c),b);#regroupement suivant b
  collect((x-a)*(x-b)*(x-c),[a,b]);#regroupement suivant a puis 
  b
  collect((x-a)*(x-b)*(x-c),[b,a]);#regroupement suivant b puis 
  a
```

$$x^3 + (-a - b - c)\,x^2 + (a b - (-a - b)\,c)\,x - a b c$$

$$(-x + b)\,(x - c)\,a + x\,(x - b)\,(x - c)$$

$$(-x + a)\,(x - c)\,b + (x - a)\,x\,(x - c)$$

$$((x - c)\,b - x\,(x - c))\,a - x\,(x - c)\,b + x^2\,(x - c)$$

$$((x - c)\,a - x\,(x - c))\,b - x\,(x - c)\,a + x^2\,(x - c)$$

L'instruction **collect** accepte en troisième argument une instruction telle que **expand**, **factor**, **normal**, **simplify** ou **distributed** qui s'applique à chacun des coefficients. L'option **distributed** signale qu'on recherche un regroupement des termes de même degré total par rapport aux différentes variables de regroupement.

```
> restart;
  collect((x-a)*(x-b)*(x-c),a,expand);#les coefficients de a
```

```
sont développés
collect((x-a)*(x-b)*(x-c),a,factor);#les coefficients de a
sont factorisés
collect((x-a)*(x-b)*(x-c),a,simplify);#Maple cherche une
simplication des coefficients
collect((x-a)*(x-b)*(x-c),[b,a],distributed);#le coefficient
de b est développé, puis le coefficient de a est développé
```

$$\left(-x^2+x\,c+b\,x-b\,c\right)a+x^3-x^2\,c-b\,x^2+x\,b\,c$$

$$-\left(-x+b\right)\left(-x+c\right)a+x\left(-x+b\right)\left(-x+c\right)$$

$$-\left(-x+b\right)\left(-x+c\right)a+x\left(-x+b\right)\left(-x+c\right)$$

$$\left(x-c\right)b\,a-x\left(x-c\right)b-x\left(x-c\right)a+x^2\left(x-c\right)$$

Ordonnancement suivant les puissances décroissantes : `sort( )`

Un polynôme **`Poly`** étant donné sous forme de somme de monômes, la commande **`sort(Poly,x)`** permet d'ordonner les termes suivant les puissances décroissantes de la variable *x*. S'il n'y a pas d'ambiguïté, **`sort`** a pour seul argument le polynôme étudié. Si le polynôme a plusieurs variables, on précise en second argument quelle est la variable à ordonner.

```
> sort(3*x^2+2.5*x^5-6/5*x^3+45);#la seule variable est x
```

$$2.5\,x^5-\frac{6}{5}\,x^3+3\,x^2+45$$

```
> sort(y^5*x^2+2.5*y^3*x^5-6/5*x^3*y^6+45*y^4,x);
  #ordonnancement selon la variable x
  sort(y^5*x^2+2.5*y^3*x^5-6/5*x^3*y^6+45*y^4,y);
  #ordonnancement selon la variable a
```

$$2.5\,y^3\,x^5-\frac{6}{5}\,y^6\,x^3+y^5\,x^2+45\,y^4$$

$$-\frac{6}{5}\,x^3\,y^6+x^2\,y^5+45\,y^4+2.5\,x^5\,y^3$$

Manipulations de fractions rationnelles

Rappelons qu'une fraction rationnelle est un quotient de polynômes. Les questions essentielles sont le plus souvent de la simplifier et de la décomposer en éléments simples.

Extraction du numérateur et du dénominateur : `numer( )`; `denom( )`

Soit $F(x)=\frac{N(x)}{M(x)}$ où $N(x)$ et $M(x)$ sont des expressions polynômiales. Les commandes **`numer(F(x))`** et **`denom(M(x))`** renvoient respectivement le numérateur et le dénominateur de $F(x)$.

```
> restart;
  F:=(x^3-5)/(x^3-7);
```

```
N:=numer(F);
M:=denom(F);#remarque : ne jamais utiliser D (opérateur Maple de dérivation) comme label d'un dénominateur!
```

$$F := \frac{x^3-5}{x^3-7}$$

$$N := x^3-5$$

$$M := x^3-7$$

Développer suivant le numérateur : `expand( )`

Soit $F(x) = \dfrac{N(x)}{M(x)}$ où $N(x)$ et $M(x)$ sont des polynômes. La commande `expand(F(x))` développe le numérateur et laisse inchangé le dénominateur.

```
> F:=(x-5)^2/((x^3-7)*(x+8));
  expand(F);
```

$$F := \frac{(x-5)^2}{(x^3-7)\,(x+8)}$$

$$\frac{x^2}{(x^3-7)\,(x+8)} - \frac{10\,x}{(x^3-7)\,(x+8)} + \frac{25}{(x^3-7)\,(x+8)}$$

Mise sous forme irréductible d'une fraction rationnelle à coefficients rationnels : `normal( )`

Soit $F(x) = \dfrac{N(x)}{\mathrm{D}(x)}$ où $N(x)$ et $\mathrm{D}(x)$ sont tous deux des polynômes **à coefficients rationnels**. La commande `normal(F(x))` renvoie un représentant irréductible de $F(x)$, c'est à dire un représentant $k\dfrac{P(x)}{Q(x)}$ dans lequel k est un nombre rationnel et *P(x)* et *Q(x)* sont des polynômes premiers entre eux.

```
> F:=(x^4-y^4)/(x^2-y^2);
  normal(F);
  F:=(x^3-(7/5)^3)/((x-7/5)*(x-2/3));
  normal(F);
```

$$F := \frac{x^4-y^4}{x^2-y^2}$$

$$x^2+y^2$$

$$F := \frac{x^3-\frac{343}{125}}{\left(x-\frac{7}{5}\right)\left(x-\frac{2}{3}\right)}$$

$$\frac{3}{25}\;\frac{25\,x^2+35\,x+49}{3\,x-2}$$

Insistons sur le fait que cette instruction n'est valable que si tous les coefficients sont rationnels. On en veut pour preuve l'exemple suivant, que Maple refuse de simplifier :

```
> F:=(x^2-2)/(x-sqrt(2));
  normal(F);
```

$$F := \frac{x^2-2}{x-\sqrt{2}}$$

$$-\frac{x^2-2}{-x+\sqrt{2}}$$

Heureusement, il y a une parade ...

Mise sous forme irréductible d'une fraction rationnelle à coefficients quelconques : `factor( )`

Soit $F(x) = \dfrac{N(x)}{M(x)}$ où $N(x)$ et $M(x)$ sont tous deux des polynômes **à coefficients quelconques**.
La commande `factor(F(x))` renvoie un représentant irréductible de $F(x)$, c'est à dire un représentant $k\dfrac{P(x)}{Q(x)}$ dans lequel k est un nombre rationnel et *P(x)* et *Q(x)* sont des produits de polynômes irréductibles **dans le corps engendré par les coefficients de *F(x)*.**

```
> F:=(x^2-2)/(x-sqrt(2));
  factor(F);
```

$$F := \frac{x^2-2}{x-\sqrt{2}}$$

$$x+\sqrt{2}$$

Encore faut-il signaler qu'il faut travailler dans le corps adéquat. Dans l'exemple suivant, tous les coefficients sont des rationnels et Maple ne trouve pas de factorisation pour le dénominateur :

```
> F:=1/(x^4-x^2+1);
  factor(F);
```

$$F := \frac{1}{x^4-x^2+1}$$

$$\frac{1}{x^4-x^2+1}$$

Il faut "forcer" la factorisation :

```
> P:=factor(F,RootOf(denom(F)));
```

$$P := 1\Big/\Big(\big(-x-\mathit{RootOf}(\_Z^4-\_Z^2+1)+\mathit{RootOf}(\_Z^4-\_Z^2+1)^3\big)\big(x-\mathit{RootOf}(\_Z^4-\_Z^2+1)+\mathit{RootOf}(\_Z^4-\_Z^2+1)^3\big)\big(x+\mathit{RootOf}(\_Z^4-\_Z^2+1)\big)\big(-x+\mathit{RootOf}(\_Z^4-\_Z^2+1)\big)\Big)$$

Et comme le résultat n'est guère exploitable, on utilise la commande `convert` avec l'option `radical` :

```
> convert(P,radical);
```

$$1\Big/\left(\left(\left(-x-\frac{1}{2}\sqrt{3}-\frac{1}{2}\,\mathrm{I}+\left(\frac{1}{2}\sqrt{3}+\frac{1}{2}\,\mathrm{I}\right)^{3}\right)\left(x-\frac{1}{2}\sqrt{3}-\frac{1}{2}\,\mathrm{I}+\left(\frac{1}{2}\sqrt{3}+\frac{1}{2}\,\mathrm{I}\right)^{3}\right)\left(x+\frac{1}{2}\sqrt{3}+\frac{1}{2}\,\mathrm{I}\right)\left(-x+\frac{1}{2}\sqrt{3}+\frac{1}{2}\,\mathrm{I}\right)\right)\right)$$

Décomposition en éléments simples : **convert**

Dans le calcul d'intégrales de fractions rationnelles, il est d'usage de commencer par décomposer l'intégrande en éléments simples. La fonction **convert** avec l'option **parfrac** a précisément pour objet de réaliser cette décomposition.

Si $F(x)$ est l'expression d'une fraction rationnelle à coefficients rationnels, alors **convert(F, parfrac,x)** retourne la décomposition de F en éléments simples **dans le corps des rationnels**.

```
> F:=(x^3+1)/(x*(x^4+4));
  convert(F,parfrac,x);
  G:=1/(x^8-1);
  convert(G,parfrac,x);
```

$$F := \frac{x^3+1}{x\,(x^4+4)}$$

$$\frac{1}{4\,x}+\frac{1}{8}\,\frac{-1-3\,x}{x^2+2\,x+2}+\frac{1}{8}\,\frac{1+x}{x^2-2\,x+2}$$

$$G := \frac{1}{x^8-1}$$

$$-\frac{1}{2\,(x^4+1)}-\frac{1}{4\,(x^2+1)}-\frac{1}{8\,(1+x)}+\frac{1}{8\,(-1+x)}$$

Si on veut décomposer $F(x)$ **dans le corps des réels ou des complexes**, il faut procéder en deux temps : utiliser la fonction **factor** avec un second paramètre permettant d'obtenir la factorisation du dénominateur et du numérateur, puis appeler **convert** :

```
> F:=(x^2+1)/(x^4-4);#expression à décomposer
  FF:=factor(F,sqrt(2));#factorisation dans le corps des réels
  convert(FF,parfrac,x);#décomposition en élements simples
  GG:=factor(F,{sqrt(2),I});#factorisation dans le corps des
  complexes
  convert(GG,parfrac,x);#décomposition en élements simples
```

$$F := \frac{x^2+1}{x^4-4}$$

$$FF := -\frac{x^2+1}{(x^2+2)\,(x+\sqrt{2})\,(-x+\sqrt{2})}$$

$$-\frac{3}{16}\,\frac{\sqrt{2}}{x+\sqrt{2}}+\frac{3}{16}\,\frac{\sqrt{2}}{x-\sqrt{2}}+\frac{1}{4\,(x^2+2)}$$

$$GG := -\frac{(-x+I)\,(x+I)}{\left(-x+I\sqrt{2}\right)\left(x+I\sqrt{2}\right)\left(x+\sqrt{2}\right)\left(-x+\sqrt{2}\right)}$$

$$-\frac{\frac{1}{16}\,I\sqrt{2}}{x-I\sqrt{2}}+\frac{\frac{1}{16}\,I\sqrt{2}}{x+I\sqrt{2}}-\frac{3}{16}\,\frac{\sqrt{2}}{x+\sqrt{2}}+\frac{3}{16}\,\frac{\sqrt{2}}{x-\sqrt{2}}$$